NAME : ***answers***

Open notes & computer, but please do not ask questions.
Write all answers on these sheets.

question	1	2	3	4	5	6	7	total
points								
maximum	15	15	15	10	20	15	10	100

1. Give the Maple commands for the operations below.

   (a) Compute a list of consecutive rational approximations of π, accurate from 2 to 21 digits. Call the list `L`.

   ```
   L := [seq(convert(evalf(Pi,n),rational,n),n=2..21)];
   ```

   (b) Remove duplicate elements of the list `L`. Of the 20 numbers in the list, how many remain after removing duplicates?

   ```
   L := [op({op(L)})]; nops(L);
                                        15
   ```

   (c) Factor the numerator and denominator of the rational numbers in the list into primes. For example: 22/7 is factored into (2)(11)/(7).

   ```
   map(t -> ifactor(numer(t))/ifactor(denom(t)),L);
   ```

/15

2. Consider the point $(1, 1)$ on the curve $f(x, y) = x^2 - y^3 - x + y = 0$.

   (a) Give the Maple command(s) to compute a Taylor series about the point $(1, 1)$ where the term of the error is of second order.

   (b) Compute the slope of the tangent line of the curve at the point $(1, 1)$ and use the slope to determine the tangent line. Write the equation of the tangent line.

   Verify that the tangent line equals the first-order Taylor series at the point $(1, 1)$.

   Answer to (a):

```
t2 := taylor(convert(taylor(f,x=1,2),polynom),y=1,2);
p := convert(t2,polynom);
```

   or `t := mtaylor(f,[x=1,y=1],2);`

   Answer to (b):

```
s := implicitdiff(f,y,x);
slope := subs(x=1,y=1,s);
L := (y - 1) = slope*(x-1);
                             L := y-1 = (1/2)*x-1/2
```

   verification: `algsubs(L,t);`

/15

3. Consider the curve defined by the equation $f(x, y) = (x^2+y^2)^5-16x^2y^2(x^2-y^2)^2 = 0$.

   (a) Give the command to make a plot of this curve, for x and y in the range from -1 to $+1$. How many points do you need to obtain a good plot?

```
plots[implicitplot](f,x=-1..1,y=-1..1,numpoints=100000);
```

   (b) Give the commands to transform this curve into polar coordinates and to plot the curve. How many times does the curve pass through (0,0)?

```
pf := subs(x=r*cos(t),y=r*sin(t),f);
s := solve(pf,r);
plots[polarplot](s[9],t=0..2*Pi);
```

```
We see 0 eight times in the solution s and indeed,
the curve passes eight times through (0,0).
```

/15

4. What is a remember table in Maple?
Explain its use and how you can create a remember table.

```
A remember table stores the results of previous calls to a Maple procedure.

We use it to make a recursion more efficient.

Adding the "option remember" to a procedure creates a remember table.
```

/10

5. The kth Lagrange polynomial in z for n points $x_1, x_2, \ldots, x_n$ is defined as

$$L[x,k,n](z) = \frac{(z-x_1)\cdots(z-x_{k-1})(z-x_{k+1})\cdots(z-x_n)}{(x_k-x_1)\cdots(x_k-x_{k-1})(x_k-x_{k+1})\cdots(x_k-x_n)}$$

So $L[x,k,n](x[k]) = 1$ and $L[x,k,n](x[i]) = 0$ for all $i \neq k$. Write a procedure `L` which returns $L[x,k,n](z)$. For example: $L[x,3,4]$ returns $\frac{(z-x_1)(z-x_2)(z-x_4)}{(x_3-x_1)(x_3-x_2)(x_3-x_4)}$.

The argument of `L` is the symbol for the independent variable of the polynomial returned by `L`. Parameters to `L` are the symbol x, and numbers k and n.

```
L := proc(z)
   local s,x,k,n,r,i:
   s := op(procname):
   x := s[1]; k := s[2]; n := s[3]:
   r := 1:
   for i from 1 to n do
      if i <> k
       then r := r*(z-x[i])/(x[k]-x[i]):
      end if;
   end do:
   return r:
end proc:
```

/20

6. Consider the system $\begin{cases} f(x,y) = y - 5x(x-1)(x-2) = 0, \\ f(y,x) = x - 5y(y-1)(y-2) = 0. \end{cases}$

(a) Give the Maple commands to plot the curve defined by $f(x,y) = 0$ in blue and $f(y,x) = 0$ in red for x and y both in the interval $[-3,+3]$.
How many intersection points do you see?

```
pf := plots[implicitplot](f,x=-3..3,y=-3..3,numpoints=10000,color=blue):
pg := plots[implicitplot](g,x=-3..3,y=-3..3,numpoints=10000):
plots[display](pf,pg);
```

```
We see five intersection points.
```

(b) Give the Maple commands to compute a triangular form of the system.
How many complex solutions does this system have? Justify your answer.

```
Groebner[Basis]([f,g],plex(x,y));
```

```
The triangular form shows there are 9 values for y and for every
value for y, there is one corresponding x value.
So the system can have 9 complex solutions.
```

/15

7. A Hankel matrix of dimension n has first row $a[1], a[2], \ldots, a[n]$.

The jth element on the ith row of the matrix equals $a[1 + ((i+j-2) \bmod n)]$.
Give the Maple command to define a Hankel matrix H_5 of dimension 5.
How many terms does the determinant of H_5 have?

```
H := Matrix(5,5,(i,j) -> a[1+((i+j-2) mod 5)]);
LinearAlgebra[Determinant](H);
nops(%);
```

26

/10